Manifold tutorial

This notebook provides a short introduction to differentiable manifolds in SageMath. The tools described below have been implemented through the SageManifolds (http://sagemanifolds.obspm.fr) project.

Click here (https://raw.githubusercontent.com/sagemanifolds/SageManifolds/master/Notebooks/SM_tutorial.ipynb) to download the notebook file (ipynb format). To run it, you must start SageMath with the Jupyter notebook, via the command `sage -n jupyter`

The following assumes that you are using version 7.5 (or higher) of SageMath:

Entrée [1]:
```
version()
```

Out[1]:
```
'SageMath version 8.8, Release Date: 2019-06-26'
```

First we set up the notebook to display mathematical objects using LaTeX rendering:

Entrée [2]:
```
%display latex
```

Defining a manifold

As an example let us define a differentiable manifold of dimension 3 over $\mathbb{R}$:

Entrée [3]:
```
M = Manifold(3, 'M', latex_name=r'\mathcal{M}', start_index=1)
```

- The first argument, `3`, is the manifold dimension. In SageManifolds, it can be any positive integer.
- The second argument, `'M'`, is a string defining the manifold's name; it may be different from the symbol set on the left-hand side of the = sign (here `M`): the latter stands for a mere Python variable, which refers to the manifold object in the computer memory, while the string `'M'` is the mathematical symbol chosen for the manifold.
- The optional argument `latex_name=r'\mathcal{M}'` sets the LaTeX symbol to display the manifold. Note the letter 'r' in front on the first quote: it indicates that the string is a *raw* one, so that the backslash character in `\mathcal` is considered as an ordinary character (otherwise, the backslash is used to escape some special characters). If the argument `latex_name` is not provided by the user, it is set to the string used as the second argument (here `'M'`)
- The optional argument `start_index=1` defines the range of indices to be used for tensor components on the manifold: setting it to 1 means that indices will range in $\{1, 2, 3\}$. The default value is `start_index=0`.

Note that the default base field is $\mathbb{R}$. If we would have used the optional argument `field='complex'`, we would have defined a manifold over $\mathbb{C}$. See the list of all options (http://doc.sagemath.org/html/en/reference/manifolds/sage/manifolds/manifold.html#sage.manifolds.manifold.Manifold) for more details.

If we ask for M, it is displayed via its LaTeX symbol:

Entrée [4]:
```
M
```

Out[4]: $\mathcal{M}$

If we use the `print` function instead, we get a short description of the object:

Entrée [5]:
```
print(M)
```

```
3-dimensional differentiable manifold M
```

Via the command `type`, we get the type of the Python object corresponding to M (here the Python class `DifferentiableManifold_with_category`):

Entrée [6]:
```
type(M)
```

Out[6]:
```
<class 'sage.manifolds.differentiable.manifold.DifferentiableManifold_with_category
```

We can also ask for the category of M and see that it is the category of smooth manifolds over $\mathbb{R}$:

Entrée [7]:
```
category(M)
```

Out[7]: $\mathbf{Smooth}_{\mathbf{R}}$

The indices on the manifold are generated by the method `irange()`, to be used in loops:

Entrée [8]:
```
for i in M.irange():
    print(i)
```

```
1
2
3
```

If the parameter `start_index` had not been specified, the default range of the indices would have been $\{0, 1, 2\}$ instead:

Entrée [9]:
```
M0 = Manifold(3, 'M', latex_name=r'\mathcal{M}')
for i in M0.irange():
    print(i)
```

```
0
1
2
```

Defining a chart on the manifold

Let us assume that the manifold $\mathcal{M}$ can be covered by a single chart (other cases are discussed below); the chart is declared as follows:

Entrée [10]:
```
X.<x,y,z> = M.chart()
```

The writing `.<x,y,z>` in the left-hand side means that the Python variables `x`, `y` and `z` are set to the three coordinates of the chart. This allows one to refer subsequently to the coordinates by their names.

In this example, the function `chart()` has no arguments, which implies that the coordinate symbols will be `x`, `y` and `z` (i.e. exactly the characters set in the `<...>` operator) and that each coordinate range is $(-\infty, +\infty)$. For other cases, an argument must be passed to `chart()` to specify the coordinate symbols and range, as well as the LaTeX symbol of a coordinate if the latter is different from the coordinate name (an example will be provided below).

The chart is displayed as a pair formed by the open set covered by it (here the whole manifold) and the coordinates:

Entrée [11]:
```
print(X)
```

```
Chart (M, (x, y, z))
```

Entrée [12]:
```
X
```

Out[12]: $(\mathcal{M}, (x, y, z))$

The coordinates can be accessed individually, by means of their indices, following the convention defined by `start_index=1` in the manifold's definition:

Entrée [13]:
```
X[1]
```

Out[13]: x

Entrée [14]:
```
X[2]
```

Out[14]: y

Entrée [15]:
```
X[3]
```

Out[15]: z

The full set of coordinates is obtained by means of the operator [:]:

Entrée [16]:
```
X[:]
```

Out[16]: (x, y, z)

Thanks to the operator `<x,y,z>` used in the chart declaration, each coordinate can be accessed directly via its name:

Entrée [17]:
```
z is X[3]
```

Out[17]: True

Coordinates are SageMath symbolic expressions:

Entrée [18]:
```
type(z)
```

Out[18]:
```
<type 'sage.symbolic.expression.Expression'>
```

Functions of the chart coordinates

Real-valued functions of the chart coordinates (mathematically speaking, *functions defined on the chart codomain*) are generated via the method `function()` acting on the chart:

Entrée [19]:
```
f = X.function(x+y^2+z^3)
f
```

Out[19]: $z^3 + y^2 + x$

Entrée [20]:
```
f.display()
```

Out[20]: $(x, y, z) \mapsto z^3 + y^2 + x$

Entrée [21]:
```
f(1,2,3)
```

Out[21]: 32

They belong to SageManifolds class `ChartFunction`:

Entrée [22]:
```
type(f)
```

Out[22]:
```
<class 'sage.manifolds.chart_func.ChartFunctionRing_with_category.element_class'>
```

and differ from SageMath standard symbolic functions by automatic simplifications in all operations. For instance, adding the two symbolic functions

Entrée [23]:
```
f0(x,y,z) = cos(x)^2; g0(x,y,z) = sin(x)^2
```

results in

Entrée [24]:
```
f0 + g0
```

Out[24]: $(x, y, z) \mapsto \cos(x)^2 + \sin(x)^2$

while the sum of the corresponding functions in the class `ChartFunction` is automatically simplified:

Entrée [25]:
```
f1 = X.function(cos(x)^2); g1 = X.function(sin(x)^2)
f1 + g1
```

Out[25]: 1

To get the same output with symbolic functions, one has to invoke the method `simplify_trig()`:

Entrée [26]:
```
(f0 + g0).simplify_trig()
```

Out[26]: $(x, y, z) \mapsto 1$

Another difference regards the display; if we ask for the symbolic function f0, we get:

Entrée [27]:
```
f0
```

Out[27]: $(x, y, z) \mapsto \cos(x)^2$

while if we ask for the chart function f1, we get only the coordinate expression:

Entrée [28]:
```
f1
```

Out[28]: $\cos(x)^2$

To get an output similar to that of f0, one should call the method `display()`:

Entrée [29]:
```
f1.display()
```

Out[29]: $(x, y, z) \mapsto \cos(x)^2$

Note that the method `expr()` returns the underlying symbolic expression:

Entrée [30]:
```
f1.expr()
```

Out[30]: $\cos(x)^2$

Entrée [31]:
```
type(f1.expr())
```

Out[31]:
```
<type 'sage.symbolic.expression.Expression'>
```

Introducing a second chart on the manifold

Let us first consider an open subset of $\mathcal{M}$, for instance the complement U of the region defined by $\{y = 0, x \geq 0\}$ (note that `(y!=0, x<0)` stands for $y \neq 0$ OR $x < 0$; the condition $y \neq 0$ AND $x < 0$ would have been written `[y!=0, x<0]` instead):

Entrée [32]:
```
U = M.open_subset('U', coord_def={X: (y!=0, x<0)})
```

Let us call `X_U` the restriction of the chart `X` to the open subset U:

Entrée [33]:
```
X_U = X.restrict(U)
X_U
```

Out[33]: $(U, (x, y, z))$

We introduce another chart on U, with spherical-type coordinates (r, θ, ϕ):

Entrée [34]:
```
Y.<r,th,ph> = U.chart(r'r:(0,+oo) th:(0,pi):\theta ph:(0,2*pi):\phi')
Y
```

Out[34]: $(U, (r, \theta, \phi))$

The function `chart()` has now some argument; it is a string, which contains specific LaTeX symbols, hence the prefix 'r' to it (for *raw* string). It also contains the coordinate ranges, since they are different from the default value, which is $(-\infty, +\infty)$. For a given coordinate, the various fields are separated by the character ':' and a space character separates the coordinates. Note that for the coordinate r, there are only two fields, since the LaTeX symbol has not to be specified. The LaTeX symbols are used for the outputs:

Entrée [35]:
```
th, ph
```

Out[35]: (θ, ϕ)

Entrée [36]:
```
Y[2], Y[3]
```
Out[36]: (θ, ϕ)

The declared coordinate ranges are now known to Sage, as we may check by means of the command `assumptions()`:

Entrée [37]:
```
assumptions()
```
Out[37]: $[\mathtt{x\ is\ real}, \mathtt{y\ is\ real}, \mathtt{z\ is\ real}, \mathtt{r\ is\ real}, r > 0, \mathtt{th\ is\ real}, \theta > 0, \theta < \pi, \mathtt{ph\ is\ real},$

They are used in simplifications:

Entrée [38]:
```
simplify(abs(r))
```
Out[38]: r

Entrée [39]:
```
simplify(abs(x)) # no simplification occurs since x can take any value in R
```
Out[39]: $|x|$

After having been declared, the chart Y can be fully specified by its relation to the chart X_U, via a transition map:

Entrée [40]:
```
transit_Y_to_X = Y.transition_map(X_U, [r*sin(th)*cos(ph), r*sin(th)*sin(ph), r*cos(th)])
transit_Y_to_X
```
Out[40]: $(U, (r, \theta, \phi)) \to (U, (x, y, z))$

Entrée [41]:
```
transit_Y_to_X.display()
```
Out[41]:
$$\begin{cases} x &= r\cos(\phi)\sin(\theta) \\ y &= r\sin(\phi)\sin(\theta) \\ z &= r\cos(\theta) \end{cases}$$

The inverse of the transition map can be specified by means of the method `set_inverse()`:

Entrée [42]:
```
transit_Y_to_X.set_inverse(sqrt(x^2+y^2+z^2), atan2(sqrt(x^2+y^2),z), atan2(y, x))
transit_Y_to_X.inverse().display()
```
Out[42]:
$$\begin{cases} r &= \sqrt{x^2 + y^2 + z^2} \\ \theta &= \arctan\left(\sqrt{x^2 + y^2}, z\right) \\ \phi &= \arctan(y, x) \end{cases}$$

At this stage, the manifold's **atlas** (the "user atlas", not the maximal atlas!) contains three charts:

Entrée [43]:
```
M.atlas()
```
Out[43]: $[(\mathcal{M}, (x, y, z)), (U, (x, y, z)), (U, (r, \theta, \phi))]$

The first chart defined on the manifold is considered as the manifold's default chart (it can be changed by the method `set_default_chart()`):

Entrée [44]:
```
M.default_chart()
```
Out[44]: $(\mathcal{M}, (x, y, z))$

Each open subset has its own atlas (since an open subset of a manifold is a manifold by itself):

Entrée [45]:
```
U.atlas()
```
Out[45]: $[(U, (x, y, z)), (U, (r, \theta, \phi))]$

Entrée [46]:
```
U.default_chart()
```

Out[46]: $(U, (x, y, z))$

We can draw the chart Y in terms of the chart X. Let us first define a viewer for 3D plots (use `'threejs'` or `'jmol'` for interactive 3D graphics):

Entrée [47]:
```
viewer3D = 'threejs' # must be 'threejs', 'jmol', 'tachyon' or None (default)
```

The plot shows lines of constant coordinates from the Y chart in a "Cartesian frame" based on the X coordinates:

Entrée [48]:
```
graph = Y.plot(X)
show(graph, viewer=viewer3D)
```



The command plot() allows for many options, to control the number of coordinate lines to be drawn, their style and color, as well as the coordinate ranges (cf. the list of all options (http://doc.sagemath.org/html/en/reference/manifolds/sage/manifolds/chart.html#sage.manifolds.chart.RealChart.plot)):

Entrée [49]:
```
graph = Y.plot(X, ranges={r:(1,2), th:(0,pi/2)}, number_values=4,
               color={r:'blue', th:'green', ph:'red'})
show(graph, aspect_ratio=1, viewer=viewer3D)
```



Conversly, the chart $X|_U$ can be plotted in terms of the chart Y (this is not possible for the whole chart X since its domain is larger than that of chart Y):

Entrée [50]:
```
graph = X_U.plot(Y)
show(graph, viewer=viewer3D, axes_labels=['r','theta','phi'])
```



Points on the manifold

A point on $\mathcal{M}$ is defined by its coordinates in a given chart:

Entrée [51]:
```
p = M.point((1,2,-1), chart=X, name='p')
print(p)
p
```

```
Point p on the 3-dimensional differentiable manifold M
```

Out[51]: p

Since $X = (\mathcal{M}, (x, y, z))$ is the manifold's default chart, its name can be omitted:

Entrée [52]:
```
p = M.point((1,2,-1), name='p')
print(p)
p
```

```
Point p on the 3-dimensional differentiable manifold M
```

Out[52]: p

Of course, p belongs to $\mathcal{M}$:

Entrée [53]:
```
p in M
```

Out[53]: True

It is also in U:

Entrée [54]:
```
p in U
```

Out[54]: True

Indeed the coordinates of p have $y \neq 0$:

Entrée [55]:
```
p.coord(X)
```

Out[55]: $(1, 2, -1)$

Note in passing that since X is the default chart on $\mathcal{M}$, its name can be omitted in the arguments of coord():

Entrée [56]:
```
p.coord()
```
Out[56]: $(1, 2, -1)$

The coordinates of p can also be obtained by letting the chart acting of the point (from the very definition of a chart!):

Entrée [57]:
```
X(p)
```
Out[57]: $(1, 2, -1)$

Let q be a point with $y = 0$ and $x \geq 0$:

Entrée [58]:
```
q = M.point((1,0,2), name='q')
```

This time, the point does not belong to U:

Entrée [59]:
```
q in U
```
Out[59]: False

Accordingly, we cannot ask for the coordinates of q in the chart $Y = (U, (r, \theta, \phi))$:

Entrée [60]:
```
try:
    q.coord(Y)
except ValueError as exc:
    print("Error: " + str(exc))
```
```
Error: the point does not belong to the domain of Chart (U, (r, th, ph))
```

but we can for point p:

Entrée [61]:
```
p.coord(Y)
```
Out[61]: $\left(\sqrt{3}\sqrt{2}, \pi - \arctan\left(\sqrt{5}\right), \arctan(2)\right)$

Entrée [62]:
```
Y(p)
```
Out[62]: $\left(\sqrt{3}\sqrt{2}, \pi - \arctan\left(\sqrt{5}\right), \arctan(2)\right)$

Points can be compared:

Entrée [63]:
```
q == p
```
Out[63]: False

Entrée [64]:
```
p1 = U.point((sqrt(3)*sqrt(2), pi-atan(sqrt(5)), atan(2)), chart=Y)
p1 == p
```
Out[64]: True

In SageMath's terminology, points are **elements**, whose **parents** are the manifold on which they have been defined:

Entrée [65]:
```
p.parent()
```
Out[65]: $\mathcal{M}$

Entrée [66]:
```
q.parent()
```
Out[66]: $\mathcal{M}$

Entrée [67]:
```
p1.parent()
```
Out[67]: U

Scalar fields

A scalar field is a differentiable mapping $U \longrightarrow \mathbb{R}$, where U is an open subset of $\mathcal{M}$.

The scalar field is defined by its expressions in terms of charts covering its domain (in general more than one chart is necessary to cover all the domain):

Entrée [68]:
```
f = U.scalar_field({X_U: x+y^2+z^3}, name='f')
print(f)
```

```
Scalar field f on the Open subset U of the 3-dimensional differentiable manifold M
```

The coordinate expressions of the scalar field are passed as a Python dictionary, with the charts as keys, hence the writing `{X_U: x+y^2+z^3}`.

Since in the present case, there is only one chart in the dictionary, an alternative writing is

Entrée [69]:
```
f = U.scalar_field(x+y^2+z^3, chart=X_U, name='f')
print(f)
```

```
Scalar field f on the Open subset U of the 3-dimensional differentiable manifold M
```

Since X_U is the domain's default chart, it can be omitted in the above declaration:

Entrée [70]:
```
f = U.scalar_field(x+y^2+z^3, name='f')
print(f)
```

```
Scalar field f on the Open subset U of the 3-dimensional differentiable manifold M
```

As a mapping $U \subset \mathcal{M} \longrightarrow \mathbb{R}$, a scalar field acts on points, not on coordinates:

Entrée [71]:
```
f(p)
```

Out[71]: 4

The method `display()` provides the expression of the scalar field in terms of a given chart:

Entrée [72]:
```
f.display(X_U)
```

Out[72]:
$$\begin{aligned} f : \ U & \longrightarrow \mathbb{R} \\ (x, y, z) & \longmapsto z^3 + y^2 + x \end{aligned}$$

If no argument is provided, the method `display()` shows the coordinate expression of the scalar field in all the charts defined on the domain (except for *subcharts*, i.e. the restrictions of some chart to a subdomain):

Entrée [73]:
```
f.display()
```

Out[73]:
$$\begin{aligned} f : \ U & \longrightarrow \mathbb{R} \\ (x, y, z) & \longmapsto z^3 + y^2 + x \\ (r, \theta, \phi) & \longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta) \end{aligned}$$

Note that the expression of f in terms of the coordinates (r, θ, ϕ) has not been provided by the user but has been automatically computed by means of the change-of-coordinate formula declared above in the transition map.

Entrée [74]:
```
f.display(Y)
```

Out[74]:
$$\begin{aligned} f : \ U & \longrightarrow \mathbb{R} \\ (r, \theta, \phi) & \longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta) \end{aligned}$$

In each chart, the scalar field is represented by a function of the chart coordinates (an object of the type `CoordFunctionSymb` described above), which is accessible via the method `coord_function()`:

Entrée [75]:
```
f.coord_function(X_U)
```
Out[75]: $z^3 + y^2 + x$

Entrée [76]:
```
f.coord_function(X_U).display()
```
Out[76]: $(x, y, z) \mapsto z^3 + y^2 + x$

Entrée [77]:
```
f.coord_function(Y)
```
Out[77]: $r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta)$

Entrée [78]:
```
f.coord_function(Y).display()
```
Out[78]: $(r, \theta, \phi) \mapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta)$

The "raw" symbolic expression is returned by the method `expr()`:

Entrée [79]:
```
f.expr(X_U)
```
Out[79]: $z^3 + y^2 + x$

Entrée [80]:
```
f.expr(Y)
```
Out[80]: $r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta)$

Entrée [81]:
```
f.expr(Y) is f.coord_function(Y).expr()
```
Out[81]: True

A scalar field can also be defined by some unspecified function of the coordinates:

Entrée [82]:
```
h = U.scalar_field(function('H')(x, y, z), name='h')
print(h)
```
```
Scalar field h on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [83]:
```
h.display()
```
Out[83]:
$$\begin{array}{rccl} h: & U & \longrightarrow & \mathbb{R} \\ & (x, y, z) & \longmapsto & H(x, y, z) \\ & (r, \theta, \phi) & \longmapsto & H(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\theta), r\cos(\theta)) \end{array}$$

Entrée [84]:
```
h.display(Y)
```
Out[84]:
$$\begin{array}{rccl} h: & U & \longrightarrow & \mathbb{R} \\ & (r, \theta, \phi) & \longmapsto & H(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\theta), r\cos(\theta)) \end{array}$$

Entrée [85]:
```
h(p) # remember that p is the point of coordinates (1,2,-1) in the chart X_U
```
Out[85]: $H(1, 2, -1)$

The parent of f is the set $C^\infty(U)$ of all smooth scalar fields on U, which is a commutative algebra over $\mathbb{R}$:

Entrée [86]:
```
CU = f.parent()
CU
```
Out[86]: $C^\infty(U)$

Entrée [87]:
```
print(CU)
```
```
Algebra of differentiable scalar fields on the Open subset U of the 3-dimensional differe
ntiable manifold M
```

Entrée [88]:
```
CU.category()
```

Out[88]: $\mathbf{CommutativeAlgebras}_{\mathrm{SR}}$

The base ring of the algebra is the field $\mathbb{R}$, which is represented here by SageMath's Symbolic Ring (SR):

Entrée [89]:
```
CU.base_ring()
```

Out[89]: SR

Arithmetic operations on scalar fields are defined through the algebra structure:

Entrée [90]:
```
s = f + 2*h
print(s)
```

```
Scalar field on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [91]:
```
s.display()
```

Out[91]:
$$\begin{aligned} U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto z^3 + y^2 + x + 2\,H\,(x, y, z) \\ (r, \theta, \phi) &\longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta) + 2\,H\,(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin \end{aligned}$$

Tangent spaces

The tangent vector space to the manifold at point p is obtained as follows:

Entrée [92]:
```
Tp = M.tangent_space(p)
Tp
```

Out[92]: $T_p\,\mathcal{M}$

Entrée [93]:
```
print(Tp)
```

```
Tangent space at Point p on the 3-dimensional differentiable manifold M
```

$T_p\,\mathcal{M}$ is a 2-dimensional vector space over $\mathbb{R}$ (represented here by SageMath's Symbolic Ring (SR)) :

Entrée [94]:
```
print(Tp.category())
```

```
Category of finite dimensional vector spaces over Symbolic Ring
```

Entrée [95]:
```
Tp.dim()
```

Out[95]: 3

$T_p\,\mathcal{M}$ is automatically endowed with vector bases deduced from the vector frames defined around the point:

Entrée [96]:
```
Tp.bases()
```

Out[96]: $\left[\left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right), \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right]$

For the tangent space at the point q, on the contrary, there is only one pre-defined basis, since q is not in the domain U of the frame associated with coordinates (r, θ, ϕ):

Entrée [97]:
```
Tq = M.tangent_space(q)
Tq.bases()
```

Out[97]: $\left[\left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right]$

A random element:

Entrée [98]:
```
v = Tp.an_element()
print(v)
```

```
Tangent vector at Point p on the 3-dimensional differentiable manifold M
```

Entrée [99]:
```
v.display()
```

Out[99]: $\frac{\partial}{\partial x} + 2\frac{\partial}{\partial y} + 3\frac{\partial}{\partial z}$

Entrée [100]:
```
u = Tq.an_element()
print(u)
```

```
Tangent vector at Point q on the 3-dimensional differentiable manifold M
```

Entrée [101]:
```
u.display()
```

Out[101]: $\frac{\partial}{\partial x} + 2\frac{\partial}{\partial y} + 3\frac{\partial}{\partial z}$

Note that, despite what the above simplified writing may suggest (the mention of the point p or q is omitted in the basis vectors), u and v are different vectors, for they belong to different vector spaces:

Entrée [102]:
```
v.parent()
```

Out[102]: $T_p\,\mathcal{M}$

Entrée [103]:
```
u.parent()
```

Out[103]: $T_q\,\mathcal{M}$

In particular, it is not possible to add u and v:

Entrée [104]:
```
try:
    s = u + v
except TypeError as exc:
    print("Error: " + str(exc))
```

```
Error: unsupported operand parent(s) for +: 'Tangent space at Point q on the 3-dimensiona
l differentiable manifold M' and 'Tangent space at Point p on the 3-dimensional different
iable manifold M'
```

Vector Fields

Each chart defines a vector frame on the chart domain: the so-called **coordinate basis**:

Entrée [105]:
```
X.frame()
```

Out[105]: $\left(\mathcal{M}, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right)$

Entrée [106]:
```
X.frame().domain()  # this frame is defined on the whole manifold
```

Out[106]: $\mathcal{M}$

Entrée [107]:
```
Y.frame()
```

Out[107]: $\left(U, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)$

Entrée [108]:
```
Y.frame().domain() # this frame is defined only on U
```

Out[108]: U

The list of frames defined on a given open subset is returned by the method `frames()`:

Entrée [109]:
```
M.frames()
```

Out[109]:
$$\left[\left(\mathcal{M}, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right), \left(U, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right), \left(U, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)\right]$$

Entrée [110]:
```
U.frames()
```

Out[110]:
$$\left[\left(U, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right), \left(U, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)\right]$$

Entrée [111]:
```
M.default_frame()
```

Out[111]:
$$\left(\mathcal{M}, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z}\right)\right)$$

Unless otherwise specified (via the command `set_default_frame()`), the default frame is that associated with the default chart:

Entrée [112]:
```
M.default_frame() is M.default_chart().frame()
```

Out[112]: True

Entrée [113]:
```
U.default_frame() is U.default_chart().frame()
```

Out[113]: True

Individual elements of a frame can be accessed by means of their indices:

Entrée [114]:
```
e = U.default_frame()
e2 = e[2]
e2
```

Out[114]:
$$\frac{\partial}{\partial y}$$

Entrée [115]:
```
print(e2)
```

```
Vector field d/dy on the Open subset U of the 3-dimensional differentiable manifold M
```

We may define a new vector field as follows:

Entrée [116]:
```
v = e[2] + 2*x*e[3]
print(v)
```

```
Vector field on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [117]:
```
v.display()
```

Out[117]:
$$\frac{\partial}{\partial y} + 2\,x\frac{\partial}{\partial z}$$

A vector field can be defined by its components with respect to a given vector frame. When the latter is not specified, the open set's default frame is of course assumed:

Entrée [118]:
```
v = U.vector_field(name='v') # vector field defined on the open set U
v[1] = 1+y
v[2] = -x
v[3] = x*y*z
v.display()
```

Out[118]:
$$v = (y+1)\,\frac{\partial}{\partial x} - x\frac{\partial}{\partial y} + xyz\frac{\partial}{\partial z}$$

Since version 8.8 of SageMath, it is possible to initialize the components of the vector field while declaring it, so that the above is equivalent to

Entrée [119]:
```
v = U.vector_field(1+y, -x, x*y*z, name='v')  # valid only in SageMath 8.8 and higher
v.display()
```

Out[119]:
$$v = (y+1)\,\frac{\partial}{\partial x} - x\frac{\partial}{\partial y} + xyz\frac{\partial}{\partial z}$$

Vector fields on U are Sage *element* objects, whose *parent* is the set $\mathfrak{X}(U)$ of vector fields defined on U:

Entrée [120]:
```
v.parent()
```

Out[120]:
$$\mathfrak{X}(U)$$

The set $\mathfrak{X}(U)$ is a module over the commutative algebra $C^\infty(U)$ of scalar fields on U:

Entrée [121]:
```
print(v.parent())
```

```
Free module X(U) of vector fields on the Open subset U of the 3-dimensional differentiabl
e manifold M
```

Entrée [122]:
```
print(v.parent().category())
```

```
Category of finite dimensional modules over Algebra of differentiable scalar fields on th
e Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [123]:
```
v.parent().base_ring()
```

Out[123]:
$$C^\infty(U)$$

A vector field acts on scalar fields:

Entrée [124]:
```
f.display()
```

Out[124]:
$$\begin{array}{rccl} f: & U & \longrightarrow & \mathbb{R} \\ & (x,y,z) & \longmapsto & z^3 + y^2 + x \\ & (r,\theta,\phi) & \longmapsto & r^3\cos(\theta)^3 + r^2\sin(\phi)^2\sin(\theta)^2 + r\cos(\phi)\sin(\theta) \end{array}$$

Entrée [125]:
```
s = v(f)
print(s)
```

```
Scalar field v(f) on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [126]:
```
s.display()
```

Out[126]:
$$\begin{array}{rccl} v(f): & U & \longrightarrow & \mathbb{R} \\ & (x,y,z) & \longmapsto & 3\,xyz^3 - (2\,x-1)y + 1 \\ & (r,\theta,\phi) & \longmapsto & -3\,r^5\cos(\phi)\cos(\theta)^5\sin(\phi) + 3\,r^5\cos(\phi)\cos(\theta)^3\sin(\phi) - 2\,r^2\cos(\phi)\sin(\phi) \end{array}$$

Entrée [127]:
```
e[3].display()
```

Out[127]:
$$\frac{\partial}{\partial z} = \frac{\partial}{\partial z}$$

Entrée [128]:
```
e[3](f).display()
```

Out[128]:
$$\begin{array}{rccl} \frac{\partial}{\partial z}(f): & U & \longrightarrow & \mathbb{R} \\ & (x,y,z) & \longmapsto & 3\,z^2 \\ & (r,\theta,\phi) & \longmapsto & 3\,r^2\cos(\theta)^2 \end{array}$$

Unset components are assumed to be zero:

Entrée [129]:
```
w = U.vector_field(name='w')
w[2] = 3
w.display()
```

Out[129]:
$$w = 3\frac{\partial}{\partial y}$$

A vector field on U can be expanded in the vector frame associated with the chart (r, θ, ϕ):

Entrée [130]:
```
v.display(Y.frame())
```

Out[130]:
$$v = \left(\frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}}\right)\frac{\partial}{\partial r} + \left(-\frac{\left(x^3y + xy^3 - x\right)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + \left(x^2 + y^2\right)z^2}\right)\frac{\partial}{\partial \theta} + \left(-\frac{x^2 + y^2 + y}{x^2 + y^2}\right)\frac{\partial}{\partial \phi}$$

By default, the components are expressed in terms of the default coordinates (x, y, z). To express them in terms of the coordinates (r, θ, ϕ), one should add the corresponding chart as the second argument of the method `display()`:

Entrée [131]:
```
v.display(Y.frame(), Y)
```

Out[131]:
$$v = \left(r^3\cos(\phi)\cos(\theta)^2\sin(\phi)\sin(\theta)^2 + \cos(\phi)\sin(\theta)\right)\frac{\partial}{\partial r} + \left(-\frac{r^3\cos(\phi)\cos(\theta)\sin(\phi)\sin(\theta)^3 - \cos(\phi}{r}\right.$$
$$\left. + \left(-\frac{r\sin(\theta) + \sin(\phi)}{r\sin(\theta)}\right)\frac{\partial}{\partial \phi}\right.$$

Entrée [132]:
```
for i in M.irange():
    show(e[i].display(Y.frame(), Y))
```

$$\frac{\partial}{\partial x} = \cos(\phi)\sin(\theta)\frac{\partial}{\partial r} + \frac{\cos(\phi)\cos(\theta)}{r}\frac{\partial}{\partial \theta} - \frac{\sin(\phi)}{r\sin(\theta)}\frac{\partial}{\partial \phi}$$
$$\frac{\partial}{\partial y} = \sin(\phi)\sin(\theta)\frac{\partial}{\partial r} + \frac{\cos(\theta)\sin(\phi)}{r}\frac{\partial}{\partial \theta} + \frac{\cos(\phi)}{r\sin(\theta)}\frac{\partial}{\partial \phi}$$
$$\frac{\partial}{\partial z} = \cos(\theta)\frac{\partial}{\partial r} - \frac{\sin(\theta)}{r}\frac{\partial}{\partial \theta}$$

The components of a tensor field w.r.t. the default frame can also be obtained as a list, via the command `[:]`:

Entrée [133]:
```
v[:]
```

Out[133]:
$$[y + 1, -x, xyz]$$

An alternative is to use the method `display_comp()`:

Entrée [134]:
```
v.display_comp()
```

Out[134]:
$$\begin{aligned} v^x &= y + 1 \\ v^y &= -x \\ v^z &= xyz \end{aligned}$$

To obtain the components w.r.t. to another frame, one may go through the method `comp()` and specify the frame:

Entrée [135]:
```
v.comp(Y.frame())[:]
```

Out[135]:
$$\left[\frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}}, -\frac{\left(x^3y + xy^3 - x\right)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + \left(x^2 + y^2\right)z^2}, -\frac{x^2 + y^2 + y}{x^2 + y^2}\right]$$

However a shortcut is to provide the frame as the first argument of the square brackets:

Entrée [136]:
```
v[Y.frame(), :]
```

Out[136]:
$$\left[\frac{xyz^2+x}{\sqrt{x^2+y^2+z^2}}, -\frac{\left(x^3y+xy^3-x\right)\sqrt{x^2+y^2}z}{x^4+2x^2y^2+y^4+\left(x^2+y^2\right)z^2}, -\frac{x^2+y^2+y}{x^2+y^2}\right]$$

Entrée [137]:
```
v.display_comp(Y.frame())
```

Out[137]:
$$\begin{aligned} v^r &= \frac{xyz^2+x}{\sqrt{x^2+y^2+z^2}} \\ v^\theta &= -\frac{\left(x^3y+xy^3-x\right)\sqrt{x^2+y^2}z}{x^4+2x^2y^2+y^4+\left(x^2+y^2\right)z^2} \\ v^\phi &= -\frac{x^2+y^2+y}{x^2+y^2} \end{aligned}$$

Components are shown expressed in terms of the default's coordinates; to get them in terms of the coordinates (r, θ, ϕ) instead, add the chart name as the last argument in the square brackets:

Entrée [138]:
```
v[Y.frame(), :, Y]
```

Out[138]:
$$\left[r^3\cos(\phi)\cos(\theta)^2\sin(\phi)\sin(\theta)^2+\cos(\phi)\sin(\theta), -\frac{r^3\cos(\phi)\cos(\theta)\sin(\phi)\sin(\theta)^3-\cos(\phi)\cos(\theta)}{r}, -\frac{r}{-}\right.$$

or specify the chart in `display_comp()`:

Entrée [139]:
```
v.display_comp(Y.frame(), chart=Y)
```

Out[139]:
$$\begin{aligned} v^r &= r^3\cos(\phi)\cos(\theta)^2\sin(\phi)\sin(\theta)^2+\cos(\phi)\sin(\theta) \\ v^\theta &= -\frac{r^3\cos(\phi)\cos(\theta)\sin(\phi)\sin(\theta)^3-\cos(\phi)\cos(\theta)}{r} \\ v^\phi &= -\frac{r\sin(\theta)+\sin(\phi)}{r\sin(\theta)} \end{aligned}$$

To get some vector component as a scalar field instead of a coordinate expression, use double square brackets:

Entrée [140]:
```
print(v[[1]])
```

```
Scalar field on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [141]:
```
v[[1]].display()
```

Out[141]:
$$\begin{aligned} U &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto y+1 \\ (r,\theta,\phi) &\longmapsto r\sin(\phi)\sin(\theta)+1 \end{aligned}$$

Entrée [142]:
```
v[[1]].expr(X_U)
```

Out[142]:
$$y+1$$

A vector field can be defined with components being unspecified functions of the coordinates:

Entrée [143]:
```
u = U.vector_field(name='u')
u[:] = [function('u_x')(x,y,z), function('u_y')(x,y,z), function('u_z')(x,y,z)]
u.display()
```

Out[143]:
$$u = u_x\left(x,y,z\right)\frac{\partial}{\partial x} + u_y\left(x,y,z\right)\frac{\partial}{\partial y} + u_z\left(x,y,z\right)\frac{\partial}{\partial z}$$

Entrée [144]:
```
s = v + u
s.set_name('s')
s.display()
```

Out[144]:
$$s = \left(y+u_x\left(x,y,z\right)+1\right)\frac{\partial}{\partial x} + \left(-x+u_y\left(x,y,z\right)\right)\frac{\partial}{\partial y} + \left(xyz+u_z\left(x,y,z\right)\right)\frac{\partial}{\partial z}$$

Values of vector fields at a given point

The value of a vector field at some point of the manifold is obtained via the method `at()`:

Entrée [145]:
```
vp = v.at(p)
print(vp)
```

```
Tangent vector v at Point p on the 3-dimensional differentiable manifold M
```

Entrée [146]:
```
vp.display()
```

Out[146]:
$$v = 3\frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2\frac{\partial}{\partial z}$$

Indeed, recall that, w.r.t. chart X_U=(x, y, z), the coordinates of the point p and the components of the vector field v are

Entrée [147]:
```
p.coord(X_U)
```

Out[147]:
$$(1, 2, -1)$$

Entrée [148]:
```
v.display(X_U.frame(), X_U)
```

Out[148]:
$$v = (y+1)\frac{\partial}{\partial x} - x\frac{\partial}{\partial y} + xyz\frac{\partial}{\partial z}$$

Note that to simplify the writing, the symbol used to denote the value of the vector field at point p is the same as that of the vector field itself (namely v); this can be changed by the method `set_name()`:

Entrée [149]:
```
vp.set_name(latex_name='v|_p')
vp.display()
```

Out[149]:
$$v|_p = 3\frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2\frac{\partial}{\partial z}$$

Of course, $v|_p$ belongs to the tangent space at p:

Entrée [150]:
```
vp.parent()
```

Out[150]:
$$T_p\,\mathcal{M}$$

Entrée [151]:
```
vp in M.tangent_space(p)
```

Out[151]:
$$\text{True}$$

Entrée [152]:
```
up = u.at(p)
print(up)
```

```
Tangent vector u at Point p on the 3-dimensional differentiable manifold M
```

Entrée [153]:
```
up.display()
```

Out[153]:
$$u = u_x\,(1,2,-1)\frac{\partial}{\partial x} + u_y\,(1,2,-1)\frac{\partial}{\partial y} + u_z\,(1,2,-1)\frac{\partial}{\partial z}$$

1-forms

A 1-form on $\mathcal{M}$ is a field of linear forms. For instance, it can be the **differential of a scalar field**:

Entrée [154]:
```
df = f.differential()
print(df)
```

```
1-form df on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [155]:
```
df.display()
```

Out[155]:
$$\mathrm{d}f = \mathrm{d}x + 2\,y\mathrm{d}y + 3\,z^2\mathrm{d}z$$

In the above writing, the 1-form is expanded over the basis $(\mathrm{d}x, \mathrm{d}y, \mathrm{d}z)$ associated with the chart (x, y, z). This basis can be accessed via the method `coframe()`:

Entrée [156]:
```
dX = X.coframe()
dX
```

Out[156]: $(\mathcal{M}, (\mathrm{d}x, \mathrm{d}y, \mathrm{d}z))$

The list of all coframes defined on a given manifold open subset is returned by the method `coframes()`:

Entrée [157]:
```
M.coframes()
```

Out[157]: $[(\mathcal{M}, (\mathrm{d}x, \mathrm{d}y, \mathrm{d}z)), (U, (\mathrm{d}x, \mathrm{d}y, \mathrm{d}z)), (U, (\mathrm{d}r, \mathrm{d}\theta, \mathrm{d}\phi))]$

As for a vector field, the value of the differential form at some point on the manifold is obtained by the method `at()`:

Entrée [158]:
```
dfp = df.at(p)
print(dfp)
```

```
Linear form df on the Tangent space at Point p on the 3-dimensional differentiable manifo
ld M
```

Entrée [159]:
```
dfp.display()
```

Out[159]: $\mathrm{d}f = \mathrm{d}x + 4\mathrm{d}y + 3\mathrm{d}z$

Recall that

Entrée [160]:
```
p.coord()
```

Out[160]: $(1, 2, -1)$

The linear form $\mathrm{d}f|_p$ belongs to the dual of the tangent vector space at p:

Entrée [161]:
```
dfp.parent()
```

Out[161]: $T_p\,\mathcal{M}^*$

Entrée [162]:
```
dfp.parent() is M.tangent_space(p).dual()
```

Out[162]: True

As such, it is acting on vectors at p, yielding a real number:

Entrée [163]:
```
print(vp)
vp.display()
```

```
Tangent vector v at Point p on the 3-dimensional differentiable manifold M
```

Out[163]: $v|_p = 3\frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2\frac{\partial}{\partial z}$

Entrée [164]:
```
dfp(vp)
```

Out[164]: -7

Entrée [165]:
```
print(up)
up.display()
```

```
Tangent vector u at Point p on the 3-dimensional differentiable manifold M
```

Out[165]: $u = u_x\,(1,2,-1)\,\frac{\partial}{\partial x} + u_y\,(1,2,-1)\,\frac{\partial}{\partial y} + u_z\,(1,2,-1)\,\frac{\partial}{\partial z}$

Entrée [166]:
```
dfp(up)
```

Out[166]: $u_x\,(1,2,-1) + 4\,u_y\,(1,2,-1) + 3\,u_z\,(1,2,-1)$

The differential 1-form of the unspecified scalar field h:

Entrée [167]:
```
dh = h.differential()
dh.display()
```

Out[167]:
$$\mathrm{d}h = \frac{\partial H}{\partial x}\mathrm{d}x + \frac{\partial H}{\partial y}\mathrm{d}y + \frac{\partial H}{\partial z}\mathrm{d}z$$

A 1-form can also be defined from scratch:

Entrée [168]:
```
om = U.one_form(name='omega', latex_name=r'\omega')
print(om)
```

```
1-form omega on the Open subset U of the 3-dimensional differentiable manifold M
```

It can be specified by providing its components in a given coframe:

Entrée [169]:
```
om[:] = [x^2+y^2, z, x-z]    # components in the default coframe (dx,dy,dz)
om.display()
```

Out[169]:
$$\omega = \left(x^2 + y^2\right) \mathrm{d}x + z\mathrm{d}y + (x - z)\,\mathrm{d}z$$

Since version 8.8 of SageMath, it is possible to initialize the components of the 1-form while declaring it, so that the above is equivalent to

Entrée [170]:
```
om = U.one_form(x^2+y^2, z, x-z, name='omega',  # valid only in
                latex_name=r'\omega')           # SageMath 8.8 and higher
om.display()
```

Out[170]:
$$\omega = \left(x^2 + y^2\right) \mathrm{d}x + z\mathrm{d}y + (x - z)\,\mathrm{d}z$$

Of course, one may set the components in a frame different from the default one:

Entrée [171]:
```
om[Y.frame(), :, Y] = [r*sin(th)*cos(ph), 0, r*sin(th)*sin(ph)]
om.display(Y.frame(), Y)
```

Out[171]:
$$\omega = r\cos(\phi)\sin(\theta)\mathrm{d}r + r\sin(\phi)\sin(\theta)\mathrm{d}\phi$$

The components in the coframe $(\mathrm{d}x, \mathrm{d}y, \mathrm{d}z)$ are updated automatically:

Entrée [172]:
```
om.display()
```

Out[172]:
$$\omega = \left(\frac{x^4 + x^2y^2 - \sqrt{x^2 + y^2 + z^2}y^2}{\sqrt{x^2 + y^2 + z^2}\left(x^2 + y^2\right)}\right)\mathrm{d}x + \left(\frac{x^3y + xy^3 + \sqrt{x^2 + y^2 + z^2}xy}{\sqrt{x^2 + y^2 + z^2}\left(x^2 + y^2\right)}\right)\mathrm{d}y + \left(\frac{xz}{\sqrt{x^2 + y^2 + z}}\right.$$

Let us revert to the values set previously:

Entrée [173]:
```
om[:] = [x^2+y^2, z, x-z]
om.display()
```

Out[173]:
$$\omega = \left(x^2 + y^2\right) \mathrm{d}x + z\mathrm{d}y + (x - z)\,\mathrm{d}z$$

This time, the components in the coframe $(\mathrm{d}r, \mathrm{d}\theta, \mathrm{d}\phi)$ are those that are updated:

Entrée [174]:
```
om.display(Y.frame(), Y)
```

Out[174]:
$$\begin{aligned}\omega = &\left(r^2\cos(\phi)\sin(\theta)^3 + r(\cos(\phi) + \sin(\phi))\cos(\theta)\sin(\theta) - r\cos(\theta)^2\right)\mathrm{d}r \\ &+ \left(r^2\cos(\theta)^2\sin(\phi) + r^2\cos(\theta)\sin(\theta) + \left(r^3\cos(\phi)\cos(\theta) - r^2\cos(\phi)\right)\sin(\theta)^2\right)\mathrm{d}\theta \\ &+ \left(-r^3\sin(\phi)\sin(\theta)^3 + r^2\cos(\phi)\cos(\theta)\sin(\theta)\right)\mathrm{d}\phi\end{aligned}$$

A 1-form acts on vector fields, resulting in a scalar field:

Entrée [175]:
```
print(om(v))
om(v).display()
```

```
Scalar field omega(v) on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[175]:

$$\begin{aligned}\omega(v):\ U &\longrightarrow \mathbb{R}\\ (x,y,z) &\longmapsto -xyz^2 + x^2y + y^3 + x^2 + y^2 + \left(x^2y - x\right)z\\ (r,\theta,\phi) &\longmapsto -r^2\cos(\phi)\cos(\theta)\sin(\theta) + \left(r^4\cos(\phi)^2\cos(\theta)\sin(\phi) + r^3\sin(\phi)\right)\sin(\theta)^3\end{aligned}$$

Entrée [176]:
```
print(df(v))
df(v).display()
```

```
Scalar field df(v) on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[176]:

$$\begin{aligned}\mathrm{d}f(v):\ U &\longrightarrow \mathbb{R}\\ (x,y,z) &\longmapsto 3\,xyz^3 - (2\,x-1)y + 1\\ (r,\theta,\phi) &\longmapsto r\sin(\phi)\sin(\theta) + \left(3\,r^5\cos(\phi)\cos(\theta)^3\sin(\phi) - 2\,r^2\cos(\phi)\sin(\phi)\right)\sin(\theta)^2\end{aligned}$$

Entrée [177]:
```
om(u).display()
```

Out[177]:

$$\begin{aligned}\omega(u):\ U &\longrightarrow \mathbb{R}\\ (x,y,z) &\longmapsto x^2u_x(x,y,z) + y^2u_x(x,y,z) + z\left(u_y(x,y,z) - u_z(x,y,z)\right) + xu_z(x,y,z)\\ (r,\theta,\phi) &\longmapsto r^2\sin(\theta)^2u_x(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\theta), r\cos(\theta)) + r\cos(\theta)u_y(r\cos(\phi)\,\mathrm{si}\\ &\quad + (r\cos(\phi)\sin(\theta) - r\cos(\theta))u_z(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\end{aligned}$$

In the case of a differential 1-form, the following identity holds:

Entrée [178]:
```
df(v) == v(f)
```

Out[178]: True

1-forms are Sage *element* objects, whose *parent* is the $C^\infty(U)$-module $\Omega^1(U)$ of all 1-forms defined on U:

Entrée [179]:
```
df.parent()
```

Out[179]: $\Omega^1(U)$

Entrée [180]:
```
print(df.parent())
```

```
Free module Omega^1(U) of 1-forms on the Open subset U of the 3-dimensional differentiabl
e manifold M
```

Entrée [181]:
```
print(om.parent())
```

```
Free module Omega^1(U) of 1-forms on the Open subset U of the 3-dimensional differentiabl
e manifold M
```

$\Omega^1(U)$ is actually the dual of the free module $\mathfrak{X}(U)$:

Entrée [182]:
```
df.parent() is v.parent().dual()
```

Out[182]: True

Differential forms and exterior calculus

The **exterior product** of two 1-forms is taken via the method `wedge()` and results in a 2-form:

Entrée [183]:
```
a = om.wedge(df)
print(a)
a.display()
```

```
2-form omega/\df on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[183]:
$$\omega \wedge \mathrm{d}f = \left(2\,x^2 y + 2\,y^3 - z\right) \mathrm{d}x \wedge \mathrm{d}y + \left(3\left(x^2 + y^2\right)z^2 - x + z\right) \mathrm{d}x \wedge \mathrm{d}z + \left(3\,z^3 - 2\,xy + 2\,yz\right) \mathrm{d}y \wedge \mathrm{d}z$$

A matrix view of the components:

Entrée [184]:
```
a[:]
```

Out[184]:
$$\begin{pmatrix} 0 & 2\,x^2 y + 2\,y^3 - z & 3\left(x^2 + y^2\right)z^2 - x + z \\ -2\,x^2 y - 2\,y^3 + z & 0 & 3\,z^3 - 2\,xy + 2\,yz \\ -3\left(x^2 + y^2\right)z^2 + x - z & -3\,z^3 + 2\,xy - 2\,yz & 0 \end{pmatrix}$$

Displaying only the non-vanishing components, skipping the redundant ones (i.e. those that can be deduced by antisymmetry):

Entrée [185]:
```
a.display_comp(only_nonredundant=True)
```

Out[185]:
$$\begin{aligned} \omega \wedge \mathrm{d}f_{x\,y} &= 2\,x^2 y + 2\,y^3 - z \\ \omega \wedge \mathrm{d}f_{x\,z} &= 3\left(x^2 + y^2\right)z^2 - x + z \\ \omega \wedge \mathrm{d}f_{y\,z} &= 3\,z^3 - 2\,xy + 2\,yz \end{aligned}$$

The 2-form $\omega \wedge \mathrm{d}f$ can be expanded on the $(\mathrm{d}r, \mathrm{d}\theta, \mathrm{d}\phi)$ coframe:

Entrée [186]:
```
a.display(Y.frame(), Y)
```

Out[186]:
$$\begin{aligned} \omega \wedge \mathrm{d}f = {} & \left(3\,r^5\cos(\phi)\sin(\theta)^4 - \left(3\,r^5\cos(\phi) - 3\,r^4\cos(\theta)\sin(\phi) - 2\,r^3\cos(\phi)\sin(\phi)^2\right)\sin(\theta)^2 - (\right. \\ & \left.(\theta) - \left(2\,r^3\cos(\theta)\sin(\phi)^2 + \left(\sin(\phi)^2 - 1\right)r^2\right)\sin(\theta)\right)\mathrm{d}r \wedge \mathrm{d}\theta \\ & + \left(2\,r^4\sin(\phi)\sin(\theta)^5 + \left(3\,r^5\cos(\theta)^3\sin(\phi) + 2\,r^3\cos(\phi)^2\cos(\theta)\sin(\phi)\right)\mathrm{si}\right. \\ & - \left(2\,r^3\cos(\phi)\cos(\theta)^2\sin(\phi) + (\cos(\phi)\sin(\phi) + 1)r^2\cos(\theta)\right)\sin(\theta)^2 - \left(3\,r^4\cos(\phi)\cos(\theta)^4 - r^2\,\mathrm{co}\right. \\ & + \left(-r^3\cos(\theta)^2\sin(\theta) - \left(3\,r^6\cos(\theta)^2\sin(\phi) + 2\,r^4\cos(\phi)^2\sin(\phi) - 2\,r^5\cos(\theta)\,\mathrm{sin}\right.\right. \\ & + \left(2\,r^4\cos(\phi)\cos(\theta)\sin(\phi) + r^3\cos(\phi)\sin(\phi)\right)\sin(\theta)^3 + \left(3\,r^5\cos(\phi)\cos(\theta)^3 - r^3\cos(\theta)\,\mathrm{sin}\right. \end{aligned}$$

As a 2-form, $A := \omega \wedge \mathrm{d}f$ can be applied to a pair of vectors and is antisymmetric:

Entrée [187]:
```
a.set_name('A')
print(a(u,v))
a(u,v).display()
```

```
Scalar field A(u,v) on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[187]:

$$
\begin{aligned}
A\,(u, v) : \quad U &\longrightarrow \mathbb{R} \\
(x, y, z) &\longmapsto 3\,xyz^4 u_y\,(x, y, z) - 2\,x^2y^2u_y\,(x, y, z) - 2\,y^4u_y\,(x, y, \\
&\quad \left(x^3yu_x\,(x, y, z) + xy^3u_x\,(x, y, \right. \\
&\quad - \left(3\,y^3u_z\,(x, y, z) - \left(2\,xu_y\,(x, y, z) - 3\,u_z\,(x, y, z)\right)y^2 + 3\,x^2\right. \\
&\quad - \left(2\,x^3u_x\,(x, y, z) + 2\,x^2u_y\,(x, y, z)\right. \\
&\quad - \left(2\,x^2y^2u_y\,(x, y, z) + \left(x^2u_x\,(x, y, z) - (2\,x - 1)u_z\,(x, y, z) - u_y\,(x, y, z\right.\right. \\
(r, \theta, \phi) &\longmapsto \left(r^4\cos(\phi)\cos(\theta)^2\sin(\phi)\sin(\theta)^2 + \left(\sin(\phi)^3 - \sin(\phi)\right)r^4\cos(\theta)\sin(\theta\right. \\
&\quad + \left(3\,r^7\cos(\phi)\cos(\theta)^3\sin(\phi) - 2\,r^4\cos(\phi)\sin(\phi)\right)\sin(\theta)^4\Big) \\
&\quad r\cos(\theta)) \\
&\quad + \left(3\,r^6\cos(\phi)\cos(\theta)^4\sin(\phi)\sin(\theta)^2 + r^2\cos(\theta)\sin(\phi)\sin(\theta) + 2\,\left(\right.\right. \\
&\quad \left(r^5\cos(\phi)\cos(\theta)^2\sin(\phi)^2 - r^3\sin(\phi)\right)\sin(\theta)^3 + r\cos(\theta)\Big) \\
&\quad (r\cos(\phi)\sin(\theta), r\sin(\phi)\,s \\
&\quad - \left(\left(3\,r^5\cos(\theta)^2\sin(\phi) - 2\left(\sin(\phi)^3 - \sin(\phi)\right)r^3\right)\sin(\theta)^3 + \left(3\,r^4\cos(\theta)^\right.\right. \\
&\quad + r\cos(\theta) - \left(3\,r^4\cos(\phi)\cos(\theta)^3 - r^2\cos(\theta)\sin(\phi) + r\cos(\phi)\right)\sin(\theta)\Big) \\
&\quad (\phi)\sin(\theta), r\sin(\phi)\sin(
\end{aligned}
$$

Entrée [188]:
```
a(u,v) == - a(v,u)
```

Out[188]: True

Entrée [189]:
```
a.symmetries()
```

```
no symmetry; antisymmetry: (0, 1)
```

The **exterior derivative** of a differential form:

Entrée [190]:
```
dom = om.exterior_derivative()
print(dom)
dom.display()
```

```
2-form domega on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[190]: $\mathrm{d}\omega = -2\,y\mathrm{d}x\wedge\mathrm{d}y + \mathrm{d}x\wedge\mathrm{d}z - \mathrm{d}y\wedge\mathrm{d}z$

Instead of invoking the method `exterior_derivative()`, one can use the function `xder`, after having imported it from `sage.manifolds.utilities`:

Entrée [191]:
```
from sage.manifolds.utilities import xder
dom = xder(om)
```

Entrée [192]:
```
da = xder(a)
print(da)
da.display()
```

```
3-form dA on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[192]: $\mathrm{d}A = \left(-6\,yz^2 - 2\,y - 1\right)\mathrm{d}x\wedge\mathrm{d}y\wedge\mathrm{d}z$

The exterior derivative is nilpotent:

Entrée [193]:
```
ddf = xder(df)
ddf.display()
```

Out[193]: $\mathrm{dd}f = 0$

Entrée [194]:
```
ddom = xder(dom)
ddom.display()
```

Out[194]: $\mathrm{dd}\omega = 0$

Lie derivative

The Lie derivative of any tensor field with respect to a vector field is computed by the method `lie_derivative()`, with the vector field as the argument:

Entrée [195]:
```
lv_om = om.lie_derivative(v)
print(lv_om)
lv_om.display()
```

```
1-form on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[195]: $\left(-yz^2 + (xy-1)z + 2\,x\right)\mathrm{d}x + \left(-xz^2 + x^2 + y^2 + \left(x^2+xy\right)z\right)\mathrm{d}y + \left(-2\,xyz + \left(x^2+1\right)y + 1\right)\mathrm{d}z$

Entrée [196]:
```
lu_dh = dh.lie_derivative(u)
print(lu_dh)
lu_dh.display()
```

```
1-form on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[196]:
$$\left(u_x\,(x,y,z)\,\frac{\partial^2 H}{\partial x^2} + u_y\,(x,y,z)\,\frac{\partial^2 H}{\partial x\partial y} + u_z\,(x,y,z)\,\frac{\partial^2 H}{\partial x\partial z} + \frac{\partial H}{\partial x}\frac{\partial u_x}{\partial x} + \frac{\partial H}{\partial y}\frac{\partial u_y}{\partial x} + \frac{\partial H}{\partial z}\frac{\partial u_z}{\partial x}\right)\mathrm{d}x$$
$$+\left(u_x\,(x,y,z)\,\frac{\partial^2 H}{\partial x\partial y} + u_y\,(x,y,z)\,\frac{\partial^2 H}{\partial y^2} + u_z\,(x,y,z)\,\frac{\partial^2 H}{\partial y\partial z} + \frac{\partial H}{\partial x}\frac{\partial u_x}{\partial y} + \frac{\partial H}{\partial y}\frac{\partial u_y}{\partial y} + \frac{\partial H}{\partial z}\frac{\partial u_z}{\partial y}\right)\mathrm{d}y$$
$$+\left(u_x\,(x,y,z)\,\frac{\partial^2 H}{\partial x\partial z} + u_y\,(x,y,z)\,\frac{\partial^2 H}{\partial y\partial z} + u_z\,(x,y,z)\,\frac{\partial^2 H}{\partial z^2} + \frac{\partial H}{\partial x}\frac{\partial u_x}{\partial z} + \frac{\partial H}{\partial y}\frac{\partial u_y}{\partial z} + \frac{\partial H}{\partial z}\frac{\partial u_z}{\partial z}\right)\mathrm{d}z$$

Let us check **Cartan identity** on the 1-form ω:

$$\mathcal{L}_v\omega = v\cdot\mathrm{d}\omega + \mathrm{d}\langle\omega, v\rangle$$

and on the 2-form A:

$$\mathcal{L}_v A = v\cdot\mathrm{d}A + \mathrm{d}(v\cdot A)$$

Entrée [197]:
```
om.lie_derivative(v) == v.contract(xder(om)) + xder(om(v))
```

Out[197]: True

Entrée [198]:
```
a.lie_derivative(v) == v.contract(xder(a)) + xder(v.contract(a))
```

Out[198]: True

The Lie derivative of a vector field along another one is the **commutator** of the two vectors fields:

Entrée [199]:
```
v.lie_derivative(u)(f) == u(v(f)) - v(u(f))
```

Out[199]: True

Tensor fields of arbitrary rank

Up to now, we have encountered tensor fields

- of type (0,0) (i.e. scalar fields),
- of type (1,0) (i.e. vector fields),
- of type (0,1) (i.e. 1-forms),
- of type (0,2) and antisymmetric (i.e. 2-forms).

More generally, tensor fields of any type (p, q) can be introduced in SageManifolds. For instance a tensor field of type (1,2) on the open subset U is declared as follows:

Entrée [200]:
```
t = U.tensor_field(1, 2, name='T')
print(t)
```

```
Tensor field T of type (1,2) on the Open subset U of the 3-dimensional differentiable man
ifold M
```

As for vectors or 1-forms, the tensor's components with respect to the domain's default frame are set by means of square brackets:

Entrée [201]:
```
t[1,2,1] = 1 + x^2
t[3,2,1] = x*y*z
```

Unset components are zero:

Entrée [202]:
```
t.display()
```

Out[202]:
$$T = \left(x^2 + 1\right) \frac{\partial}{\partial x} \otimes \mathrm{d}y \otimes \mathrm{d}x + xyz \frac{\partial}{\partial z} \otimes \mathrm{d}y \otimes \mathrm{d}x$$

Entrée [203]:
```
t[:]
```

Out[203]:
$$\left[\left[0, 0, 0\right], \left[x^2 + 1, 0, 0\right], \left[0, 0, 0\right]\right], \left[\left[0, 0, 0\right], \left[0, 0, 0\right], \left[0, 0, 0\right]\right], \left[\left[0, 0, 0\right], \left[xyz, 0, 0\right], \left[0, 0, 0\right]\right]$$

Display of the nonzero components:

Entrée [204]:
```
t.display_comp()
```

Out[204]:
$$\begin{aligned} T^{x}{}_{y\,x} &= x^2 + 1 \\ T^{z}{}_{y\,x} &= xyz \end{aligned}$$

Double square brackets return the component (still w.r.t. the default frame) as a scalar field, while single square brackets return the expression of this scalar field in terms of the domain's default coordinates:

Entrée [205]:
```
print(t[[1,2,1]])
t[[1,2,1]].display()
```

```
Scalar field on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[205]:
$$\begin{aligned} U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto x^2 + 1 \\ (r, \theta, \phi) &\longmapsto r^2 \cos\left(\phi\right)^2 \sin\left(\theta\right)^2 + 1 \end{aligned}$$

Entrée [206]:
```
print(t[1,2,1])
t[1,2,1]
```

```
x^2 + 1
```

Out[206]:
$$x^2 + 1$$

A tensor field of type (1,2) maps a 3-tuple (1-form, vector field, vector field) to a scalar field:

Entrée [207]:
```
print(t(om, u, v))
t(om, u, v).display()
```

```
Scalar field T(omega,u,v) on the Open subset U of the 3-dimensional differentiable manifo
ld M
```

Out[207]:

$$\begin{aligned} T(\omega,u,v): \quad U &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto \left(x^2+1\right)y^3u_y(x,y,z)+\left(x^2+1\right)y^2u_y(x,y,z)-\left(xy^2u_y(x,y,\right. \\ &\quad +\left(x^2y^2u_y(x,y,z)+x^2yu_y(x,y,z)\right)z \\ (r,\theta,\phi) &\longmapsto \left(r^5\cos(\phi)^2\sin(\phi)\sin(\theta)^5-\left(\left(\cos(\phi)^4-\cos(\phi)^2\right)r^5\cos(\theta)-r^4\cos(\right.\right. \\ &\quad +\left(\left(\cos(\phi)^3-\cos(\phi)\right)r^5\cos(\theta)^2+r^4\cos(\phi)^2\cos(\theta)\sin(\phi)+r^3\sin(\right. \\ &\quad (r\cos(\phi)\sin(\theta),r\sin(\phi)\sin \end{aligned}$$

As for vectors and differential forms, the tensor components can be taken in any frame defined on the manifold:

Entrée [208]:
```
t[Y.frame(), 1,1,1, Y]
```

Out[208]:

$$r^2\cos(\phi)^4\sin(\phi)\sin(\theta)^5+\left(\cos(\phi)^4-\cos(\phi)^2\right)r^3\sin(\theta)^6-\left(\cos(\phi)^4-\cos(\phi)^2\right)r^3\sin(\theta)^4+\cos$$

Tensor calculus

The **tensor product** $\otimes$ is denoted by `*`:

Entrée [209]:
```
print(v.tensor_type())
print(a.tensor_type())
```

```
(1, 0)
(0, 2)
```

Entrée [210]:
```
b = v*a
print(b)
b
```

```
Tensor field v*A of type (1,2) on the Open subset U of the 3-dimensional differentiable m
anifold M
```

Out[210]:

$$v \otimes A$$

The tensor product preserves the (anti)symmetries: since A is a 2-form, it is antisymmetric with respect to its two arguments (positions 0 and 1); as a result, b is antisymmetric with respect to its last two arguments (positions 1 and 2):

Entrée [211]:
```
a.symmetries()
```

```
no symmetry; antisymmetry: (0, 1)
```

Entrée [212]:
```
b.symmetries()
```

```
no symmetry; antisymmetry: (1, 2)
```

Standard **tensor arithmetics** is implemented:

Entrée [213]:
```
s = - t + 2*f* b
print(s)
```

```
Tensor field of type (1,2) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Tensor contractions are dealt with by the methods `trace()` and `contract()`: for instance, let us contract the tensor T w.r.t. its first two arguments (positions 0 and 1), i.e. let us form the tensor c of components $c_i = T^k_{\ ki}$:

Entrée [214]:
```
c = t.trace(0,1)
print(c)
```

```
1-form on the Open subset U of the 3-dimensional differentiable manifold M
```

An alternative to the writing `trace(0,1)` is to use the **index notation** to denote the contraction: the indices are given in a string inside the [] operator, with '^' in front of the contravariant indices and '_' in front of the covariant ones:

Entrée [215]:
```
c1 = t['^k_ki']
print(c1)
c1 == c
```

```
1-form on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[215]: True

The contraction is performed on the repeated index (here k); the letter denoting the remaining index (here i) is arbitrary:

Entrée [216]:
```
t['^k_kj'] == c
```

Out[216]: True

Entrée [217]:
```
t['^b_ba'] == c
```

Out[217]: True

It can even be replaced by a dot:

Entrée [218]:
```
t['^k_k.'] == c
```

Out[218]: True

LaTeX notations are allowed:

Entrée [219]:
```
t['^{k}_{ki}'] == c
```

Out[219]: True

The contraction $T^i_{jk} v^k$ of the tensor fields T and v is taken as follows (2 refers to the last index position of T and 0 to the only index position of v):

Entrée [220]:
```
tv = t.contract(2, v, 0)
print(tv)
```

```
Tensor field of type (1,1) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Since 2 corresponds to the last index position of T and 0 to the first index position of v, a shortcut for the above is

Entrée [221]:
```
tv1 = t.contract(v)
print(tv1)
```

```
Tensor field of type (1,1) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Entrée [222]:
```
tv1 == tv
```

Out[222]: True

Instead of `contract()`, the **index notation**, combined with the * operator, can be used to denote the contraction:

Entrée [223]:
```
t['^i_jk']*v['^k'] == tv
```

Out[223]: True

The non-repeated indices can be replaced by dots:

Entrée [224]:
```
t['^._.k']*v['^k'] == tv
```

Out[224]: True

Metric structures

A **Riemannian metric** on the manifold $\mathcal{M}$ is declared as follows:

Entrée [225]:
```
g = M.riemannian_metric('g')
print(g)
```

```
Riemannian metric g on the 3-dimensional differentiable manifold M
```

It is a symmetric tensor field of type (0,2):

Entrée [226]:
```
g.parent()
```

Out[226]: $\mathcal{T}^{(0,2)}\left(\right)$

Entrée [227]:
```
print(g.parent())
```

```
Free module T^(0,2)(M) of type-(0,2) tensors fields on the 3-dimensional differentiable m
anifold M
```

Entrée [228]:
```
g.symmetries()
```

```
symmetry: (0, 1); no antisymmetry
```

The metric is initialized by its components with respect to some vector frame. For instance, using the default frame of $\mathcal{M}$:

Entrée [229]:
```
g[1,1], g[2,2], g[3,3] = 1, 1, 1
g.display()
```

Out[229]: $g = \mathrm{d}x \otimes \mathrm{d}x + \mathrm{d}y \otimes \mathrm{d}y + \mathrm{d}z \otimes \mathrm{d}z$

The components w.r.t. another vector frame are obtained as for any tensor field:

Entrée [230]:
```
g.display(Y.frame(), Y)
```

Out[230]: $g = \mathrm{d}r \otimes \mathrm{d}r + r^2 \mathrm{d}\theta \otimes \mathrm{d}\theta + r^2 \sin(\theta)^2 \mathrm{d}\phi \otimes \mathrm{d}\phi$

Of course, the metric acts on vector pairs:

Entrée [231]:
```
print(g(u,v))
g(u,v).display()
```

```
Scalar field g(u,v) on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[231]:
$$\begin{aligned} g(u,v): \quad U &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto xyzu_z(x,y,z) + yu_x(x,y,z) - xu_y(x,y,z) + u_x(x,y,z) \\ (r,\theta,\phi) &\longmapsto r^3\cos(\phi)\cos(\theta)\sin(\phi)\sin(\theta)^2 u_z(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\theta) \\ &\quad (\theta)u_y(r\cos(\phi)\sin(\theta), r\sin(\phi)\sin(\theta), r\cos(\theta)) + (r\sin(\phi)\sin(\theta)+1)u_x(r\cc \end{aligned}$$

The **Levi-Civita connection** associated to the metric g:

Entrée [232]:
```
nabla = g.connection()
print(nabla)
nabla
```

```
Levi-Civita connection nabla_g associated with the Riemannian metric g on the 3-dimension
al differentiable manifold M
```

Out[232]: ∇_g

The Christoffel symbols with respect to the manifold's default coordinates:

Entrée [233]:
```
nabla.coef()[:]
```

Out[233]: $[[[0, 0, 0], [0, 0, 0], [0, 0, 0]], [[0, 0, 0], [0, 0, 0], [0, 0, 0]], [[0, 0, 0], [0, 0, 0], [0, 0, 0]]]$

The Christoffel symbols with respect to the coordinates (r, θ, ϕ):

Entrée [234]:
```
nabla.coef(Y.frame())[:, Y]
```

Out[234]:
$$\left[\left[[0, 0, 0], [0, -r, 0], \left[0, 0, -r \sin(\theta)^2\right]\right], \left[\left[0, \frac{1}{r}, 0\right], \left[\frac{1}{r}, 0, 0\right], [0, 0, -\cos(\theta) \sin(\theta)]\right], \left[\left[0, 0, \frac{1}{r}\right], \left[0, 0, \frac{\cos(\theta)}{\sin(\theta)}\right], \left[\frac{1}{r}, \frac{\cos(\theta)}{\sin(\theta)}, 0\right]\right]\right]$$

A nice view is obtained via the method `display()` (by default, only the nonzero connection coefficients are shown):

Entrée [235]:
```
nabla.display(frame=Y.frame(), chart=Y)
```

Out[235]:
$$\begin{aligned}
\Gamma^{r}_{\ \theta\,\theta} &= -r \\
\Gamma^{r}_{\ \phi\,\phi} &= -r \sin(\theta)^2 \\
\Gamma^{\theta}_{\ r\,\theta} &= \frac{1}{r} \\
\Gamma^{\theta}_{\ \theta\,r} &= \frac{1}{r} \\
\Gamma^{\theta}_{\ \phi\,\phi} &= -\cos(\theta) \sin(\theta) \\
\Gamma^{\phi}_{\ r\,\phi} &= \frac{1}{r} \\
\Gamma^{\phi}_{\ \theta\,\phi} &= \frac{\cos(\theta)}{\sin(\theta)} \\
\Gamma^{\phi}_{\ \phi\,r} &= \frac{1}{r} \\
\Gamma^{\phi}_{\ \phi\,\theta} &= \frac{\cos(\theta)}{\sin(\theta)}
\end{aligned}$$

One may also use the method `christoffel_symbols_display()` of the metric, which (by default) displays only the non-redundant Christoffel symbols:

Entrée [236]:
```
g.christoffel_symbols_display(Y)
```

Out[236]:
$$\begin{aligned}
\Gamma^{r}_{\ \theta\,\theta} &= -r \\
\Gamma^{r}_{\ \phi\,\phi} &= -r \sin(\theta)^2 \\
\Gamma^{\theta}_{\ r\,\theta} &= \frac{1}{r} \\
\Gamma^{\theta}_{\ \phi\,\phi} &= -\cos(\theta) \sin(\theta) \\
\Gamma^{\phi}_{\ r\,\phi} &= \frac{1}{r} \\
\Gamma^{\phi}_{\ \theta\,\phi} &= \frac{\cos(\theta)}{\sin(\theta)}
\end{aligned}$$

The connection acting as a covariant derivative:

Entrée [237]:
```
nab_v = nabla(v)
print(nab_v)
nab_v.display()
```

```
Tensor field nabla_g(v) of type (1,1) on the Open subset U of the 3-dimensional different
iable manifold M
```

Out[237]:
$$\nabla_g v = \frac{\partial}{\partial x} \otimes \mathrm{d}y - \frac{\partial}{\partial y} \otimes \mathrm{d}x + yz \frac{\partial}{\partial z} \otimes \mathrm{d}x + xz \frac{\partial}{\partial z} \otimes \mathrm{d}y + xy \frac{\partial}{\partial z} \otimes \mathrm{d}z$$

Being a Levi-Civita connection, ∇_g is torsion.free:

Entrée [238]:
```
print(nabla.torsion())
nabla.torsion().display()
```

```
Tensor field of type (1,2) on the 3-dimensional differentiable manifold M
```

Out[238]: 0

In the present case, it is also flat:

Entrée [239]:
```
print(nabla.riemann())
nabla.riemann().display()
```

```
Tensor field Riem(g) of type (1,3) on the 3-dimensional differentiable manifold M
```

Out[239]: $\mathrm{Riem}\,(g) = 0$

Let us consider a non-flat metric, by changing g_{rr} to $1/(1+r^2)$:

Entrée [240]:
```
g[Y.frame(), 1,1, Y] = 1/(1+r^2)
g.display(Y.frame(), Y)
```

Out[240]: $$g = \left(\frac{1}{r^2+1}\right) \mathrm{d}r \otimes \mathrm{d}r + r^2 \mathrm{d}\theta \otimes \mathrm{d}\theta + r^2 \sin(\theta)^2 \mathrm{d}\phi \otimes \mathrm{d}\phi$$

For convenience, we change the default chart on the domain U to Y=$(U, (r, \theta, \phi))$:

Entrée [241]:
```
U.set_default_chart(Y)
```

In this way, we do not have to specify Y when asking for coordinate expressions in terms of (r, θ, ϕ):

Entrée [242]:
```
g.display(Y.frame())
```

Out[242]: $$g = \left(\frac{1}{r^2+1}\right) \mathrm{d}r \otimes \mathrm{d}r + r^2 \mathrm{d}\theta \otimes \mathrm{d}\theta + r^2 \sin(\theta)^2 \mathrm{d}\phi \otimes \mathrm{d}\phi$$

We recognize the metric of the hyperbolic space $\mathbb{H}^3$. Its expression in terms of the chart $(U, (x, y, z))$ is

Entrée [243]:
```
g.display(X_U.frame(), X_U)
```

Out[243]: $$g = \left(\frac{y^2+z^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}x \otimes \mathrm{d}x + \left(-\frac{xy}{x^2+y^2+z^2+1}\right) \mathrm{d}x \otimes \mathrm{d}y + \left(-\frac{xz}{x^2+y^2+z^2+1}\right) \mathrm{d}x \otimes$$
$$+ \left(-\frac{xy}{x^2+y^2+z^2+1}\right) \mathrm{d}y \otimes \mathrm{d}x + \left(\frac{x^2+z^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}y \otimes \mathrm{d}y + \left(-\frac{yz}{x^2+y^2+z^2+1}\right) \mathrm{d}y \otimes \mathrm{d}$$
$$+ \left(-\frac{xz}{x^2+y^2+z^2+1}\right) \mathrm{d}z \otimes \mathrm{d}x + \left(-\frac{yz}{x^2+y^2+z^2+1}\right) \mathrm{d}z \otimes \mathrm{d}y + \left(\frac{x^2+y^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}z \otimes \mathrm{d}$$

A matrix view of the components may be more appropriate:

Entrée [244]:
```
g[X_U.frame(), :, X_U]
```

Out[244]: $$\begin{pmatrix} \frac{y^2+z^2+1}{x^2+y^2+z^2+1} & -\frac{xy}{x^2+y^2+z^2+1} & -\frac{xz}{x^2+y^2+z^2+1} \\ -\frac{xy}{x^2+y^2+z^2+1} & \frac{x^2+z^2+1}{x^2+y^2+z^2+1} & -\frac{yz}{x^2+y^2+z^2+1} \\ -\frac{xz}{x^2+y^2+z^2+1} & -\frac{yz}{x^2+y^2+z^2+1} & \frac{x^2+y^2+1}{x^2+y^2+z^2+1} \end{pmatrix}$$

We extend these components, a priori defined only on U, to the whole manifold $\mathcal{M}$, by demanding the same coordinate expressions in the frame associated to the chart X=$(\mathcal{M}, (x, y, z))$:

Entrée [245]:
```
g.add_comp_by_continuation(X.frame(), U, X)
g.display()
```

Out[245]:

$$g = \left(\frac{y^2+z^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}x\otimes\mathrm{d}x + \left(-\frac{xy}{x^2+y^2+z^2+1}\right) \mathrm{d}x\otimes\mathrm{d}y + \left(-\frac{xz}{x^2+y^2+z^2+1}\right) \mathrm{d}x\otimes$$
$$+ \left(-\frac{xy}{x^2+y^2+z^2+1}\right) \mathrm{d}y\otimes\mathrm{d}x + \left(\frac{x^2+z^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}y\otimes\mathrm{d}y + \left(-\frac{yz}{x^2+y^2+z^2+1}\right) \mathrm{d}y\otimes\mathrm{d}$$
$$+ \left(-\frac{xz}{x^2+y^2+z^2+1}\right) \mathrm{d}z\otimes\mathrm{d}x + \left(-\frac{yz}{x^2+y^2+z^2+1}\right) \mathrm{d}z\otimes\mathrm{d}y + \left(\frac{x^2+y^2+1}{x^2+y^2+z^2+1}\right) \mathrm{d}z\otimes\mathrm{d}$$

The Levi-Civita connection is automatically recomputed, after the change in g:

Entrée [246]:
```
nabla = g.connection()
```

In particular, the Christoffel symbols are different:

Entrée [247]:
```
nabla.display(only_nonredundant=True)
```

Out[247]:

$$\Gamma^{x}{}_{x\,x} = -\frac{xy^2+xz^2+x}{x^2+y^2+z^2+1}$$
$$\Gamma^{x}{}_{x\,y} = \frac{x^2y}{x^2+y^2+z^2+1}$$
$$\Gamma^{x}{}_{x\,z} = \frac{x^2z}{x^2+y^2+z^2+1}$$
$$\Gamma^{x}{}_{y\,y} = -\frac{x^3+xz^2+x}{x^2+y^2+z^2+1}$$
$$\Gamma^{x}{}_{y\,z} = \frac{xyz}{x^2+y^2+z^2+1}$$
$$\Gamma^{x}{}_{z\,z} = -\frac{x^3+xy^2+x}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{x\,x} = -\frac{y^3+yz^2+y}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{x\,y} = \frac{xy^2}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{x\,z} = \frac{xyz}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{y\,y} = -\frac{yz^2+(x^2+1)y}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{y\,z} = \frac{y^2z}{x^2+y^2+z^2+1}$$
$$\Gamma^{y}{}_{z\,z} = -\frac{y^3+(x^2+1)y}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{x\,x} = -\frac{z^3+(y^2+1)z}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{x\,y} = \frac{xyz}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{x\,z} = \frac{xz^2}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{y\,y} = -\frac{z^3+(x^2+1)z}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{y\,z} = \frac{yz^2}{x^2+y^2+z^2+1}$$
$$\Gamma^{z}{}_{z\,z} = -\frac{(x^2+y^2+1)z}{x^2+y^2+z^2+1}$$

Entrée [248]:
```
nabla.display(frame=Y.frame(), chart=Y, only_nonredundant=True)
```

Out[248]:

$$\Gamma^{r}{}_{r\,r} = -\frac{r}{r^2+1}$$
$$\Gamma^{r}{}_{\theta\,\theta} = -r^3 - r$$
$$\Gamma^{r}{}_{\phi\,\phi} = -\left(r^3 + r\right)\sin(\theta)^2$$
$$\Gamma^{\theta}{}_{r\,\theta} = \frac{1}{r}$$
$$\Gamma^{\theta}{}_{\phi\,\phi} = -\cos(\theta)\sin(\theta)$$
$$\Gamma^{\phi}{}_{r\,\phi} = \frac{1}{r}$$
$$\Gamma^{\phi}{}_{\theta\,\phi} = \frac{\cos(\theta)}{\sin(\theta)}$$

The **Riemann tensor** is now

Entrée [249]:
```
Riem = nabla.riemann()
print(Riem)
Riem.display(Y.frame())
```

```
Tensor field Riem(g) of type (1,3) on the 3-dimensional differentiable manifold M
```

Out[249]:

$$\mathrm{Riem}(g) = -r^2\frac{\partial}{\partial r}\otimes \mathrm{d}\theta\otimes \mathrm{d}r\otimes \mathrm{d}\theta + r^2\frac{\partial}{\partial r}\otimes \mathrm{d}\theta\otimes \mathrm{d}\theta\otimes \mathrm{d}r - r^2\sin(\theta)^2\frac{\partial}{\partial r}\otimes \mathrm{d}\phi\otimes \mathrm{d}r\otimes \mathrm{d}\phi + r$$
$$\otimes \mathrm{d}r + \left(\frac{1}{r^2+1}\right)\frac{\partial}{\partial \theta}\otimes \mathrm{d}r\otimes \mathrm{d}r\otimes \mathrm{d}\theta + \left(-\frac{1}{r^2+1}\right)\frac{\partial}{\partial \theta}\otimes \mathrm{d}r\otimes \mathrm{d}\theta\otimes \mathrm{d}r - r^2\sin(\theta)^2\frac{\partial}{\partial \theta}\otimes \mathrm{d}\phi\otimes$$
$$\otimes \mathrm{d}\phi\otimes \mathrm{d}\phi\otimes \mathrm{d}\theta + \left(\frac{1}{r^2+1}\right)\frac{\partial}{\partial \phi}\otimes \mathrm{d}r\otimes \mathrm{d}r\otimes \mathrm{d}\phi + \left(-\frac{1}{r^2+1}\right)\frac{\partial}{\partial \phi}\otimes \mathrm{d}r\otimes \mathrm{d}\phi\otimes \mathrm{d}r + r^2\frac{\partial}{\partial \phi}\otimes$$
$$\otimes \mathrm{d}\theta\otimes \mathrm{d}\phi\otimes \mathrm{d}\theta$$

Note that it can be accessed directely via the metric, without any explicit mention of the connection:

Entrée [250]:
```
g.riemann() is nabla.riemann()
```

Out[250]: True

The **Ricci tensor** is

Entrée [251]:
```
Ric = g.ricci()
print(Ric)
Ric.display(Y.frame())
```

```
Field of symmetric bilinear forms Ric(g) on the 3-dimensional differentiable manifold M
```

Out[251]:

$$\mathrm{Ric}(g) = \left(-\frac{2}{r^2+1}\right)\mathrm{d}r\otimes \mathrm{d}r - 2\,r^2\mathrm{d}\theta\otimes \mathrm{d}\theta - 2\,r^2\sin(\theta)^2\mathrm{d}\phi\otimes \mathrm{d}\phi$$

The **Weyl tensor** is:

Entrée [252]:
```
C = g.weyl()
print(C)
C.display()
```

```
Tensor field C(g) of type (1,3) on the 3-dimensional differentiable manifold M
```

Out[252]: $\mathrm{C}(g) = 0$

The Weyl tensor vanishes identically because the dimension of $\mathcal{M}$ is 3.

Finally, the **Ricci scalar** is

Entrée [253]:
```
R = g.ricci_scalar()
print(R)
R.display()
```

```
Scalar field r(g) on the 3-dimensional differentiable manifold M
```

Out[253]:

$$\begin{aligned} \mathrm{r}(g): \quad \mathcal{M} &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto -6 \\ \text{on } U: \quad (r, \theta, \phi) &\longmapsto -6 \end{aligned}$$

We recover the fact that $\mathbb{H}^3$ is a Riemannian manifold of constant negative curvature.

Tensor transformations induced by a metric

The most important tensor transformation induced by the metric g is the so-called **musical isomorphism**, or **index raising** and **index lowering**:

Entrée [254]:
```
print(t)
```

```
Tensor field T of type (1,2) on the Open subset U of the 3-dimensional differentiable man
ifold M
```

Entrée [255]:
```
t.display()
```

Out[255]:

$$T = \left(r^2 \cos(\phi)^2 \sin(\theta)^2 + 1\right) \frac{\partial}{\partial x} \otimes \mathrm{d}y \otimes \mathrm{d}x + r^3 \cos(\phi)\cos(\theta)\sin(\phi)\sin(\theta)^2 \frac{\partial}{\partial z} \otimes \mathrm{d}y \otimes \mathrm{d}x$$

Entrée [256]:
```
t.display(X_U.frame(), X_U)
```

Out[256]:

$$T = \left(x^2 + 1\right) \frac{\partial}{\partial x} \otimes \mathrm{d}y \otimes \mathrm{d}x + xyz \frac{\partial}{\partial z} \otimes \mathrm{d}y \otimes \mathrm{d}x$$

Raising the last index of T with g:

Entrée [257]:
```
s = t.up(g, 2)
print(s)
```

```
Tensor field of type (2,1) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Raising all the covariant indices of T (i.e. those at the positions 1 and 2):

Entrée [258]:
```
s = t.up(g)
print(s)
```

```
Tensor field of type (3,0) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Entrée [259]:
```
s = t.down(g)
print(s)
```

```
Tensor field of type (0,3) on the Open subset U of the 3-dimensional differentiable manif
old M
```

Hodge duality

The volume 3-form (Levi-Civita tensor) associated with the metric g is

Entrée [260]:
```
epsilon = g.volume_form()
print(epsilon)
epsilon.display()
```

```
3-form eps_g on the 3-dimensional differentiable manifold M
```

Out[260]:

$$\epsilon_g = \left(\frac{1}{\sqrt{x^2 + y^2 + z^2 + 1}}\right) \mathrm{d}x \wedge \mathrm{d}y \wedge \mathrm{d}z$$

Entrée [261]:
```
epsilon.display(Y.frame())
```

Out[261]:
$$\epsilon_g = \left(\frac{r^2 \sin(\theta)}{\sqrt{r^2+1}} \right) \mathrm{d}r \wedge \mathrm{d}\theta \wedge \mathrm{d}\phi$$

Entrée [262]:
```
print(f)
f.display()
```

```
Scalar field f on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[262]:
$$\begin{aligned} f : \ U &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto z^3 + y^2 + x \\ (r,\theta,\phi) &\longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta) \end{aligned}$$

Entrée [263]:
```
sf = f.hodge_dual(g)
print(sf)
sf.display()
```

```
3-form *f on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[263]:
$$\star f = \left(\frac{r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r\cos(\phi)\sin(\theta)}{\sqrt{r^2+1}} \right) \mathrm{d}x \wedge \mathrm{d}y \wedge \mathrm{d}z$$

We check the classical formula $\star f = f\, \epsilon_g$, or, more precisely, $\star f = f\, \epsilon_g|_U$ (for f is defined on U only):

Entrée [264]:
```
sf == f * epsilon.restrict(U)
```

Out[264]:
True

The Hodge dual of a 1-form is a 2-form:

Entrée [265]:
```
print(om)
om.display()
```

```
1-form omega on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[265]:
$$\omega = r^2 \sin(\theta)^2 \mathrm{d}x + r\cos(\theta)\mathrm{d}y + (r\cos(\phi)\sin(\theta) - r\cos(\theta))\,\mathrm{d}z$$

Entrée [266]:
```
som = om.hodge_dual(g)
print(som)
som.display()
```

```
2-form *omega on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[266]:
$$\star\omega = \left(\frac{r^4 \cos(\phi)\cos(\theta)\sin(\theta)^3 - r^3\cos(\theta)^3 - r\cos(\theta) + \left(r^3(\cos(\phi)+\sin(\phi))\cos(\theta)^2 + r\cos(\phi)\right.}{\sqrt{r^2+1}} \right.$$
$$+ \left(-\frac{r^4\cos(\phi)\sin(\phi)\sin(\theta)^4 - r^3\cos(\theta)^2\sin(\phi)\sin(\theta) + \left(\cos(\phi)\sin(\phi) + \sin(\phi)^2\right)r^3\cos(\theta)\sin(\theta)^2}{\sqrt{r^2+1}} \right.$$
$$+ \left(\frac{r^4\cos(\phi)^2\sin(\theta)^4 - r^3\cos(\phi)\cos(\theta)^2\sin(\theta) + \left(\left(\cos(\phi)^2 + \cos(\phi)\sin(\phi)\right)r^3\cos(\theta) + r^2\right)\mathrm{si}}{\sqrt{r^2+1}} \right.$$

The Hodge dual of a 2-form is a 1-form:

Entrée [267]:
```
print(a)
```

```
2-form A on the Open subset U of the 3-dimensional differentiable manifold M
```

Entrée [268]:
```
sa = a.hodge_dual(g)
print(sa)
sa.display()
```

```
1-form *A on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[268]:

$$\star A = \left(\frac{\begin{array}{l} 3\,r^5\cos(\theta)^5 + 3\,r^3\cos(\theta)^3 + \left(3\,r^6\cos(\phi)\cos(\theta)^2\sin(\phi) - 2\,r^5\cos(\phi)\cos(\theta)\sin(\phi) - 2\,r^4\,c \right. \\ + \left(2\,r^4\cos(\theta)\sin(\phi)^3 + \left(\sin(\phi)^3 - \sin(\phi)\right)r^3\right)\sin(\theta)^3 \\ + \left(3\,r^5\cos(\theta)^3\sin(\phi)^2 - 2\,r^4\cos(\phi)\cos(\theta)^2\sin(\phi) + r^3\cos(\phi)\cos(\theta)\sin(\phi) - 2\,r^2\cos(\phi)\right. \\ \left. + \left(2\,r^4\cos(\theta)^3\sin(\phi) + r^3\cos(\phi)\cos(\theta)^2 + 2\,r^2\cos(\theta)\sin(\phi)\right)\sin(\theta) \right. \end{array}}{\sqrt{r^2+1}} \right.$$

$$+ \left(-\frac{\begin{array}{l} r^3\cos(\theta)^3 + \left(3\,r^6\cos(\phi)^2\cos(\theta)^2 - 2\left(\cos(\phi)^2 - 1\right)r^5\cos(\theta) + 2\left(\cos(\phi)^4 - \cos(\phi\right.\right. \\ - \left(r^3\cos(\phi)^3 + 2\left(\cos(\phi)^3 - \cos(\phi)\right)r^4\cos(\theta)\right)\sin(\theta)^3 \\ + \left(3\,r^6\cos(\theta)^4 + 3\,r^5\cos(\phi)\cos(\theta)^3\sin(\phi) + r^3\cos(\phi)^2\cos(\theta) + 3\,r^4\cos(\theta)^2\right)\sin(\epsilon \\ \left. - \left(r^3(\cos(\phi) + \sin(\phi))\cos(\theta)^2 + r\cos(\phi)\right)\sin(\theta) \right. \end{array}}{\sqrt{r^2+1}} \right.$$

$$+ \left(\frac{\begin{array}{l} 2\,r^5\sin(\phi)\sin(\theta)^5 + \left(3\,r^6\cos(\theta)^3\sin(\phi) + 2\,r^4\cos(\phi)^2\cos(\theta)\sin(\phi) + 2\,r^3\sin(\phi)\right)\sin(\theta)^3 \\ - \left(2\,r^4\cos(\phi)\cos(\theta)^2\sin(\phi) + (\cos(\phi)\sin(\phi) + 1)r^3\cos(\theta)\right)\sin(\theta)^2 - r\cos(\theta) - \left(3\,r^5\cos(\phi)\cos\right. \\ \left. (\theta) \right. \end{array}}{\sqrt{r^2+1}} \right.$$

$$\left. \vphantom{\frac{1}{1}} \right) \mathrm{d}z$$

Finally, the Hodge dual of a 3-form is a 0-form:

Entrée [269]:
```
print(da)
da.display()
```

```
3-form dA on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[269]:

$$\mathrm{d}A = \left(-2\left(3\,r^3\cos(\theta)^2\sin(\phi) + r\sin(\phi)\right)\sin(\theta) - 1\right)\mathrm{d}x \wedge \mathrm{d}y \wedge \mathrm{d}z$$

Entrée [270]:
```
sda = da.hodge_dual(g)
print(sda)
sda.display()
```

```
Scalar field *dA on the Open subset U of the 3-dimensional differentiable manifold M
```

Out[270]:

$$\begin{aligned} \star\mathrm{d}A : \quad U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto -\left(6\,yz^2 + 2\,y + 1\right)\sqrt{x^2 + y^2 + z^2 + 1} \\ (r, \theta, \phi) &\longmapsto -\sqrt{r^2 + 1}\left(2\left(3\,r^3\cos(\theta)^2\sin(\phi) + r\sin(\phi)\right)\sin(\theta) + 1\right) \end{aligned}$$

In dimension 3 and for a Riemannian metric, the Hodge star is idempotent:

Entrée [271]:
```
sf.hodge_dual(g) == f
```

Out[271]: True

Entrée [272]:
```
som.hodge_dual(g) == om
```

Out[272]: True

Entrée [273]:
```
sa.hodge_dual(g) == a
```

Out[273]: True

Entrée [274]:
```
sda.hodge_dual(g) == da
```

Out[274]: True

Getting help

To get the list of functions (methods) that can be called on a object, type the name of the object, followed by a dot and the TAB key, e.g.

```
sa.
```

To get information on an object or a method, use the question mark:

Entrée [275]:
```
nabla?
```

Entrée [276]:
```
g.ricci_scalar?
```

Using a double question mark leads directly to the **Python source code** (SageMath is **open source**, isn't it?)

Entrée [277]:
```
g.ricci_scalar??
```

Going further

Have a look at the examples on SageManifolds page (http://sagemanifolds.obspm.fr/examples.html), especially the 2-dimensional sphere example (http://nbviewer.jupyter.org/github/sagemanifolds/SageManifolds/blob/master/Notebooks/SM_sphere_S2.ipynb) for usage on a non-parallelizable manifold (each scalar field has to be defined in at least two coordinate charts, the module $\mathfrak{X}(\mathcal{M})$ is no longer free and each tensor field has to be defined in at least two vector frames).